# $Id$
# Maintainer: Jan de Groot <jgc@archlinux.org>

pkgname=atk
pkgver=2.24.0
pkgrel=2
pkgdesc="The interface definitions of accessibility infrastructure"
url="https://git.gnome.org/browse/atk"
arch=(i686 x86_64)
license=(LGPL)
depends=(glib2)
makedepends=(gobject-introspection git gtk-doc)
_commit=7d1f9a6fcf4654c41e64707d3684645ac037c239  # tags/ATK_2_24_0^0
source=("git+https://git.gnome.org/browse/atk#commit=$_commit")
sha256sums=('SKIP')

pkgver() {
  cd $pkgname
  git describe --tags | sed 's/^ATK_//;s/_/./g;s/-/+/g'
}

prepare() {
  cd $pkgname
  NOCONFIGURE=1 ./autogen.sh
}

build() {
  cd $pkgname
  ./configure --prefix=/usr --enable-gtk-doc
  make
}

package() {
  cd $pkgname
  make DESTDIR="$pkgdir" install
}
